2F

PNL-4945
UC-70

INFIL1D: A Quasi-Analytical Model for Simulating One-Dimensional, Constant Flux Infiltration

C. S. Simmons
T. J. McKeon

April 1984

Prepared for the U.S. Department of Energy
under Contract DE-AC06-76RLO 1830

Pacific Northwest Laboratory
Operated for the U.S. Department of Energy
by Battelle Memorial Institute

Battelle

PNL-4945

DISCLAIMER

This report was prepared as an account of work sponsored by an agency of the United States Government. Neither the United States Government nor any agency thereof, nor any of their employees, makes any warranty, express or implied, or assumes any legal liability or responsibility for the accuracy, completeness, or usefulness of any information, apparatus, product, or process disclosed, or represents that its use would not infringe privately owned rights. Reference herein to any specific commercial product, process, or service by trade name, trademark, manufacturer, or otherwise, does not necessarily constitute or imply its endorsement, recommendation, or favoring by the United States Government or any agency thereof. The views and opinions of authors expressed herein do not necessarily state or reflect those of the United States Government or any agency thereof.

PACIFIC NORTHWEST LABORATORY
operated by
BATTELLE
for the
UNITED STATES DEPARTMENT OF ENERGY
under Contract DE-AC06-76RLO 1830

Printed in the United States of America
Available from
National Technical Information Service
United States Department of Commerce
5285 Port Royal Road
Springfield, Virginia 22161

NTIS Price Codes
Microfiche A01

Printed Copy

Pages	Price Codes
001-025	A02
026-050	A03
051-075	A04
076-100	A05
101-125	A06
126-150	A07
151-175	A08
176-200	A09
201-225	A010
226-250	A011
251-275	A012
276-300	A013

3 3679 00058 7271

PNL-4945
UC-70

INFIL1D: A QUASI-ANALYTICAL MODEL FOR SIMULATING ONE-DIMENSIONAL, CONSTANT FLUX INFILTRATION

C. S. Simmons
T. J. McKeon

April 1984

Prepared for
the U.S. Department of Energy
under Contract DE-AC06-76RLO 1830

Pacific Northwest Laboratory
Richland, Washington 99352

ACKNOWLEDGMENTS

This work, performed by Pacific Northwest Laboratory, was supported by the Department of Energy (DOE) Low-Level Waste Management Program, under Contract DE-AC06-76RLO 1830 with the U.S. Department of Energy. Pacific Northwest Laboratory is operated by Battelle Memorial Institute for the U.S. Department of Energy.

ABSTRACT

The program INFIL1D is designed to calculate approximate wetting-front advance into an unsaturated, uniformly moist, homogeneous soil profile, under constant surface-flux conditions. The code is based on a quasi-analytical method, which utilizes an assumed invariant functional relationship between reduced (normalized) flux and water content. The code uses general hydraulic property data in tabular form to simulate constant surface-flux infiltration.

SUMMARY

A computer program called INFIL1D was developed to calculate wetting-front movement into uniform soils under constant-flux infiltration conditions. The program is based on a quasi-analytical solution to the unsaturated flow equation, as derived by Perroux, Smiles and White (1981). Philip's (1973) concept of a flux-concentration relationship for the soil-water content profile was used to obtain an approximate solution. Plug flow occurring behind a wetting front when saturation of the soil surface takes place is also simulated. Upon surface saturation, a wetting front then advances by maintaining its shape invariantly ahead of a saturated plug flow zone, which enters the soil according to soil-water conservation constraints.

Hydraulic properties (diffusivity and conductivity) for an unsaturated soil are input in tabular form, and calculations are performed with intermediate values interpolated from the input table. This allows for use of entirely general property data.

The program uses a numerical integration over the involved water content range and is free of numerical difficulties often associated with finite difference methods. Special asymptotic integration features were included to treat singularities in the involved integral equations that describe wetting-front movement. Thus, the INFIL1D program is ideal for evaluating the accuracy of numerical infiltration simulations. To verify the program, a comparison with simulations of Haverkamp et al. (1977) was performed. The program was also tested with the original data obtained by Perroux, Smiles and White (1981).

CONTENTS

FIGURES

INTRODUCTION

The calculation of detailed hourly or daily moisture balance within an unsaturated zone requires an accurate prediction of infiltration during rainfall events. Infiltration is usually estimated by numerically solving the nonlinear Richard's flow equation for wetting-front advance under either ponded or specified surface-flux boundary conditions. A specified surface-flux condition is generally the most appropriate description of a rainfall event.

Numerical control of wetting-front simulation, however, is perhaps the most difficult aspect of unsaturated zone computer modeling. The difficulty stems from the extreme changes that occur in hydraulic properties over the involved range of soil-water contents. Moreover, the difficulty is compounded because the hydraulic property dependence on water content may alter drastically with different soils.

Numerical simulations of wetting-front movement can exhibit severe oscillations that cause associated inaccuracy in profile flux, unless the equation-solving algorithm, spatial discretization, and time-step size are carefully coordinated. Therefore, accurate quasi-analytical solutions are needed, which can be employed to test, evaluate, and calibrate more general numerical codes.

Analytical and quasi-analytical solutions of Richard's equation for infiltration under constant surface head (ponding) conditions were those first extensively studied, as reviewed by Philip (1969). Philip's asymptotic series expansion in terms of the square root of time is the most well-known, quasi-analytical solution. More exact analytical solutions are essentially impossible to find unless special, over-simplified hydraulic properties are assumed (Philip 1974). Tractable simplifications usually make a solution inappropriate for actual soil conditions.

To solve the infiltration problem under constant surface-flux conditions, and to simplify the flow equation, Philip (1973) introduced the concept of a flux-concentration relationship, based on equating relative flux to a function of relative water content, independent of depth. White, Smiles and Perroux

(1979) demonstrated the validity of that concept for absorption (horizontal infiltration). A method for predicting vertical infiltration based on the flux-concentration relationship was then described concisely by Perroux, Smiles and White (1981), in conjunction with supporting experimental justification.

This report discusses the computer model called INFIL1D, which was designed to implement the quasi-analytical method presented by Perroux, Smiles and White (1981). Certain mathematical difficulties associated with singularities in the controlling integral equations, however, were not treated and accounted for by those authors. Those potential difficulties are considered in this report. Their method was further extended to include the situation when a saturated zone (plug flow) advances behind the unsaturated wetting front. Thus, surface-water content, which increases with time during infiltration, need not be restricted from reaching saturation as in the original derivation.

The INFIL1D code allows for use of completely general hydraulic properties (soil-water diffusivity and conductivity point data), and an arbitrary constant surface flux can be specified. Options for various limiting-case, flux-concentration functions are also included. Although not an exact result, the quasi-analytical solution has proven very accurate. An extensive comparison of infiltration simulations presented by Haverkamp et al. (1977) was used to further demonstrate the program.

THEORY

The partial differential equation (a nonlinear Fokker-Plank equation) that describes vertical water movement in an unsaturated soil is

$$d\theta/dt = d/dz[D(\theta)d\theta/dz - K(\theta)] \tag{1}$$

where θ is volumetric water content (L^3/L^3); $D(\theta)$ is soil-water diffusivity (L^2/t); $K(\theta)$ is hydraulic conductivity (L/t); and z is depth (L) defined as positive downward. (Here d/dt and d/dz denote partial derivatives.) Equation (1) is related to Richard's unsaturated flow equation through the definition of soil-water diffusivity

$$D(\theta) = K(\theta)dh/d\theta \tag{2}$$

where $h(\theta)$ is the soil-water pressure head (matric potential). The relationship between h and θ is called the soil-water characteristic curve (moisture retention curve). Darcy's flux, denoted $V(\theta)$, as a functional of θ, is

$$V(\theta) = -D(\theta)d\theta/dz + K(\theta) \tag{3}$$

The first term of Equation (3) is the moisture diffusion flux; the hydraulic conductivity term is the gravity component, which always moves downward.

Equation (1) is valid for all moisture-flow conditions only if $h(\theta)$ is unique and single-valued so that hysteresis behavior is excluded (i.e., Equation (1) applies to isothermal unsaturated flow in nonswelling soils). Equation (3), therefore, applies only when θ is less than the saturated value. Diffusivity is undefined (infinite) at saturated conditions. Darcy's law, when based on the pressure head gradient and hydraulic conductivity, however, remains applicable throughout all conditions. But for the single-flow process of infiltration into uniformly moist soil, Equation (1) still applies for hysteretic soils provided that the wetting-curve branch of a characteristic curve is used to define diffusivity. Also, a capillary fringe should not be present. Of course, hydraulic conductivity must be unique and the influence of soil-air movement must be negligible.

For horizontal absorption, the gravity term of Equation (3) vanishes, and Equation (1) becomes a nonlinear diffusion equation. Absorption under constant water-content (or pressure head) conditions has been extensively studied by Philip (1969). The problem of absorption and infiltration under constant surface-flux conditions was only given emphasis more recently (Philip 1974; Perroux, Smiles and White 1981).

Under field conditions, infiltration is primarily controlled by surface flux that equals the rainfall rate. Not until the soil surface becomes saturated and the water ponds does a specified surface pressure head become the appropriate boundary condition. Prasad and Romkens (1982) described a model incorporating both of these stages of infiltration. However, if a soil surface crust controls the infiltration process, a flux boundary condition may remain appropriate for the entire time. The mathematical discussion presented here will follow this latter conceptualization. Smiles, Knight and Perroux (1982) have described absorption governed by a soil crust.

In the case of constant surface flux, the ponding pressure head must become positive and increase in time when that flux exceeds saturated conductivity, Ks. Within a saturated zone, Equation (3) no longer holds, and the Darcy flux becomes

$$V(\theta s) = -Ks\ dh/dz + Ks$$

Supposing that θ does not deviate from saturation with time implies that the flux, V, is constant and so is the pressure head gradient. Pressure head, h, is zero at the saturated zone's moving interface with the unsaturated wetting front. Thus, this boundary condition is clearly unlike that for constant head ponding conditions.

QUASI-ANALYTICAL SOLUTION

The discussion is concerned with solving Equation (1) for infiltration under a constant surface-flux boundary condition. A homogeneous soil profile with initially uniform moisture content is assumed. As a consequence of the extreme nonlinearity, a general analytical solution of Equation (1) does not exist. However, Philip (1973) found an approximate, quasi-analytical method for describing infiltration by presuming existence of a flux-concentration relation as follows:

$$F(\hat{\theta},t) = (V(\theta)-Kn)/(Vo-Kn) \tag{4}$$

where $\hat{\theta} = (\theta - \theta n)/(\theta o - \theta n)$ is the reduced water content, and Kn is $K(\theta n)$ for the initial water content θn. The surface-water content, which increases with time from the beginning of infiltration under the constant flux, Vo, is $\theta o(t)$. Relevant initial and boundary conditions are

$$\theta(z,0) = \theta n, \quad z > 0$$
$$\theta(0,t) = \theta o(t), \quad t > 0$$
$$V(\theta) = Vo, \quad z = 0, \quad t > 0.$$

A function F defined by Equation (4) would depend in general on the particular hydraulic properties $D(\theta)$ and $K(\theta)$. However, certain mathematical properties would be common to all such functions. For $0 < \hat{\theta} < 1$, F would increase monotonically and satisfy $0 < \hat{\theta} \leq F(\hat{\theta},t) < 1$. Philip identified an envelope for the possible F, based on special soil properties and absorption conditions. White, Smiles and Perroux (1979) reviewed the nature of those $F(\hat{\theta},t)$ and indicated that the minimum (lower bound) is

$$F(\hat{\theta},t) = \hat{\theta} \tag{5a}$$

and the maximum (upper bound) is

$$F(\hat{\theta},t) = \sin[\frac{\pi}{2}\ \hat{\theta}^{\pi/4}] \quad (5b)$$

Equation (5a) is associated with a delta-function diffusivity at θo and Equation (5b) with a constant value, both under conditions of constant θo. An intermediate function F for constant D and Vo was found to be approximated by

$$F(\hat{\theta},t) = \hat{\theta}^{2-4/\pi} \quad (6)$$

For horizontal absorption, White, Smiles and Perroux (1979) found that the wetting-front moisture profiles were not sensitive to the particular F used when compared with the influence of uncertainty in D(θ). Perroux, Smiles and White (1981) extended the approximate solution method to vertical infiltration and suggested that Equation (5a) was adequate. A study by Haverkamp et al. (1977) also supported the adequacy of Equation (5a) for describing wetting-front advance. Calculation of the precise F by means of an iterative procedure was derived by Philip and Knight (1974). Apparently, any $F(\hat{\theta},t)$ between the lower and upper bounds given by Equations (5a) and (5b) determines a first approximation for a wetting front. Moreover, any linear combination of F functions (with positive multipliers summing to unity) is again a possible flux-concentration relation. Smiles, Knight and Perroux (1982) gave another F function that best fits actual measurements for absorption. Most importantly, however, the equations derived by Perroux, Smiles and White provide a simplified method for determining bounds on the wetting-front advance. In many cases those bounds may actually fall within experimental uncertainty and, therefore, constitute a sufficiently accurate description of wetting-front movement.

The approximate quasi-analytical method based on an assumed known $F(\hat{\theta})$, which is independent of time, is derived again below and extended to include the situation when time exceeds a finite ponding time, Tp. A finite Tp occurs only when Vo exceeds the saturated hydraulic conductivity, Ks. The derived equations are the basis for the INFIL1D program.

An equation for the wetting-front profile is obtained by substituting Equation (3) for flux into Equation (4) and by integrating for $z(\theta,t)$ as an explicit function of θ. The resulting equation is

$$(Vo - Kn)(z - Zs(t)) = \int_{\theta}^{\theta_o(t)} G(\theta,\theta o(t))\, d\theta \qquad (7)$$

where we have defined

$$G(\theta,\theta o) = D(\theta)/[F(\hat{\theta}) - (K(\theta)-Kn)/(Vo-Kn)]$$

for $\theta n<\theta\leq\theta o$. The function $Zs(t)$ is the location of maximum θ within the unsaturated zone. Letting Tp denote a finite time at which the soil surface becomes saturated, $Zs(t)$ equals zero for $t<Tp$. If $\theta o(t)$ does not attain the saturated value θs at any time, then Tp is infinite and $Zs(t)$ vanishes in Equation (7). A finite Tp exists only when $Vo>K(\theta s)$, as will be demonstrated by obtaining the time when each θo is reached.

Equations for the unknown $\theta o(t)$ and $Zs(t)$ are found by using the statement of soil-water conservation:

$$d/dt \int_{\theta_n}^{\theta} z\, d\theta = V(\theta) - Kn \qquad (8)$$

The constant surface-flux boundary condition then gives

$$\int_{\theta_n}^{\theta_o(t)} z\, d\theta = (Vo-Kn)t \qquad (9)$$

Equation (9) defines the total soil-water storage added to the profile after time t. Substitution of $z(\theta,t)$, from Equation (7), into Equation (9) and integration by parts yields:

$$(Vo-Kn)^2 t = \int_{\theta_n}^{\theta_o(t)} (\theta - \theta n) G(\theta, \theta o(t))\, d\theta \qquad (10)$$

Equation (10) determines time t when each $\theta o(t)$ is reached.

Certain mathematical properties of $G(\theta,\theta o)$ for $\theta n<\theta<\theta o\leq\theta s$ determine the behavior of wetting-front advance. Because $G(\theta,\theta o)$ has a nonintegrable singularity at θn, then $z(\theta,t) \rightarrow +\infty$ as $\theta \rightarrow \theta n$, according to Equation (7). The integrand of Equation (10), however, is integrable near θn, because both $F(\hat{\theta})$ and $[K(\theta)-Kn]$ are asymptotically proportional to $(\theta-\theta n)$ near θn. The program makes use of an asymptotic expression for $G(\theta,\theta o)$ near θn to calculate the integrals of Equations (7) and (10). The leading edge of a wetting front must therefore be calculated for a minimum water content, θ_1, set arbitrarily above θn. The profile between θ_1 and θn can be made a negligible 'tail', which constitutes an arbitrarily small storage quantity.

Two infiltration situations depending on the magnitude of Vo are identified as follows. These two cases determine the behavior of surface water content $\theta o(t)$ and depend on the Ks value.

Case Vo<Ks: There exists a maximum θo, which is called θm, such that $Vo=K(\theta m)$. $G(\theta,\theta m)$ has a nonintegrable singularity at θm, so that $z(\theta,t) \rightarrow +\infty$ for every θ when $\theta o \rightarrow \theta m$. Moreover, Tp determined by Equation (10) for $\theta o=\theta m$ is infinite. Note that θm equal to θs is not excluded as a possibility. Thus, in this case the time to reach maximum surface-water content is infinite, and the advancing front extends to infinity as that maximum is approached.

Case Vo>Ks: Now θm equals θs, but $G(\theta,\theta s)$ does not have a singularity for $\theta n<\theta\leq\theta s$. This means that a finite Tp determined by Equation (10) with $\theta o=\theta s$ exists. Therefore when $t\leq Tp$, the front location determined by Equation (7) is everywhere finite.

When $V_o > K_s$, the extent of the saturated zone $Z_s(t)$ is determined from Equation (9) for $\theta_o(t)=\theta_s$ when $t \geq T_p$. The water storage satisfies:

$$\int_{\theta_n}^{\theta_s} z \, d\theta = \int_{\theta_n}^{\theta_s} Z_s \, d\theta + \int_{\theta_n}^{\theta_s} (z - Z_s) \, d\theta \qquad (11)$$

$$= (\theta_s - \theta_n) Z_s + (V_o - K_n) T_p$$

By combining Equations (9) and (11), Zs is given by

$$Z_s(t) = (V_o - K_n)(t - T_p)/(\theta_s - \theta_n) \qquad (12)$$

for $t \geq T_p$; and Zs vanishes otherwise.

When the wetting-front movement has been determined from Equations (7) and (10), the infiltration flux for each depth and time can be found directly from the defined flux-concentration relationship of Equation (4). Accuracy of the quasi-analytical wetting front is checked in the INFIL1D program by computing the storage integral, Equation (9). Also, the program is designed to calculate horizontal absorption by removing the gravitational flow term from the Darcy flux. Resulting equations, which are simpler, were derived by White, Smiles and Perroux (1979).

For a wetting-front advance that approximates that of absorption, the water-content profiles can be scaled by introducing reduced variables:

$$Z = (V_o - K_n) z \qquad (13a)$$

$$T = (V_o - K_n)^2 \, t \qquad (13b)$$

White, Smiles and Perroux (1979) and Perroux, Smiles and White (1981) used these reduced variables.

NUMERICAL DETAILS

In order that the wetting front be well-defined, the functions $F(\hat{\theta})$ and $K(\theta)$ need to satisfy a compatibility condition:

$$F[(\theta-\theta n)/(\theta m-\theta n)] \geq (K(\theta)-Kn)/(K(\theta m)-Kn) \tag{14}$$

when θ is between θn and θm. Then

$$F(\hat{\theta}) > (K(\theta) - Kn)/(Vo - Kn) \tag{15}$$

for $\theta n<\theta\leq\theta o<\theta m$, since

$$F(\hat{\theta}) > F[(\theta-\theta n)/(\theta m-\theta n)]$$

and $Vo \geq K(\theta m)$. Inequality, Equation (14), is only a sufficient condition so that $G(\theta,\theta o)$ has no singularity for $\theta n<\theta\leq\theta o<\theta m$. The program INFIL1D checks directly that the necessary inequality, Equation (15), holds over the integration interval of Equation (10). If the inequality, Equation (14), holds at specified tabulated values of $K(\theta)$, and $K(\theta)$ is represented as a convex function over the θ intervals between the specified values, then Equation (15) will hold. A function is strictly convex when, a linear interpolation between two function values is greater than the actual intermediate function value. For instance, a linear or exponential interpolation over an interval defines a convex function. Note that all properly defined F are concave functions, and the required inequality may hold for an interpolation that is not necessarily convex.

Because Equation (10) is an implicit function for $\theta o(t)$ at each t, an iterative method is used in INFIL1D to calculate θo at specified times. Iterations are performed until θo has a time within a user-specified tolerance. Wetting-front location is then calculated for the associated approximate θo. The entire method, of course, is quasi-analytical because required integrals are numerically evaluated. An IMSL (1980) integration routine is employed by INFIL1D, but any other adequately accurate algorithm could be applied instead. To estimate required integrals, the code uses logrithmic-linear interpolation of hydraulic property values between the tabular input. This means that hydraulic properties are represented as exponential curves between the specified values.

APPLICATIONS

The INFIL1D code was compared with several other published results. The first comparison was made with results published by Haverkamp et al. (1977). Those authors present both experimental data and simulations based on a variety of finite difference numerical methods. The hydraulic properties of a sand were used [Haverkamp et al. 1977, Equation (12)]. Initial water content was 0.1 cm^3/cm^3, and the flux was 13.7 cm/hour. Simulation results shown in Figure 1 compare quite well with both the experimental and numerical results presented by Haverkamp et al. (1977, Figure 2).

The Haverkamp et al. (1977) example represents a case for which Vo is much less than saturated conductivity, Ks, equal to 34 cm/hour. For this case, the infiltration does not behave as absorption. Equation (10) gave better estimates of the surface water content than those obtained with Parlange's method, as discussed by Haverkamp et al. (1977)



FIGURE 1. Infiltration Profiles at Times: 0.1, 0.3, 0.5, and 0.8 Hours for a Sand (Haverkamp et al. 1977). Surface flux equals 13.7 cm/hour.

A second comparison was made using the results published by Perroux, Smiles and White (1981). The soil type considered was Bungendore fine sand. Hydraulic properties of the sand were taken from figures given in that paper and from the diffusivity data given earlier by White, Smiles and Perroux (1979). The sand's initial water content was 0.01 m^3/m^3. Three comparisons were made: one with a flux 2.8 x 10^{-5} m/sec, another with a flux 5.9 x 10^{-5} m/sec, and a third absorption simulation repeated with the larger flux. The results are given in Figures 2, 3, and 4 in terms of reduced depth Z and time T, Equations (13a) and (13b).

Figure 3, for flux 5.9 x 10^{-5} m/sec, shows good agreement the with results of Perroux, Smiles and White (1981, Figure 4). Those shown in Figure 2 for flux 2.8 x 10^{-5} m/sec, however, do not compare as well. The calculated front has not moved as far down as that presented by Perroux, Smiles and White



FIGURE 2. Infiltration Profiles at Reduced Times: 2.52, 8.0, 16.1 X 10^{-7} m*m/sec for a Fine Sand (Perroux, Smiles and White 1981, Figure 4). Surface flux equals 2.8 x 10^{-5} m/sec.



FIGURE 3. Infiltration Profiles at Reduced Times: 2.52, 8.0, 16.1 x 10^{-7} m*m/sec for a Fine Sand (Perroux, Smiles and White 1981, Figure 4). Surface flux equals 5.9 x 10^{-5} m/sec.

(1981), but the surface-water content and trailing edge for the wetting front show agreement. Absorption results shown in Figure 4 compare well with those of Perroux, Smiles and White (1981). These are cases for which flux is greater than or equal to saturated conductivity, and where infiltration approaches an absorption front.

A water balance based on trapezoidal integration of the water-content profile was used to check accuracy. Thus far, using only 25 depth intervals for Z, the largest error in water balance was less than 2% and was generally less than 1%. An increase in the number of intervals would likely further reduce errors in water balance.



FIGURE 4. Horizontal Absorption Profiles for Values of Figure 3

RECOMMENDATIONS

Subjects that deserve further study and some possible changes in the model are

1. Development of a self-contained integration routine to replace the IMSL (1980) routine currently used. A Simpson rule integration should be adequate.
2. Evaluation of other flux-concentration relations for more accurate wetting-front advance. It should be possible to identify parameters that give the best fit of a $F(\hat{\theta})$ function to numerically simulated fronts.
3. Reformulation of surface-flux conditions holding after ponding so that flux can vary depending on ponding depth as a function of time.

REFERENCES

Haverkamp, R., et al. 1977. "A Comparison of Numerical Simulation Models for One-Dimensional Infiltration." Soil Sci.Soc. Am. J. 41:285-294.

International Mathematical and Statistical Libraries, Inc. (IMSL). 1980. The IMSL Library Reference Manual. Edition 8. Houston, Texas.

Perroux, K. M., D. E. Smiles, and I. White. 1981. "Water Movement in Uniform Soils During Constant-Flux Infiltration." Soil Sci. Soc. Am. J. 45:237-240.

Philip, J. R. 1969. "Theory of Infiltration." Adv. Hydrosci. 5:215-305, Academic Press, N.Y.

Philip, J. R. 1973. "On Solving the Unsaturated Flow Equation: 1. The Flux-Concentration Relation." Soil Science 116(5):328-335.

Philip, J. R. 1974. "Recent Progress in the Solution of Non-linear Diffusion Equations." Soil Science 117:257-264.

Philip, J. R., and J. H. Knight. 1974. "On Solving theUnsaturated Flow Equation: 3. New Quasi-analytical Technique." Soil Science 117:1-13.

Prasad, S. N., and M. J. M. Romkens. 1982. "An Approximate Integral Solution of Vertical Infiltration Under Changing Boundary Conditions." Water Resour. Res. 18:1022-1028.

Smiles, D. E., J. H. Knight and K. M. Perroux. 1982. "Absorption of Water by Soil: The Effect of a Surface Crust." Soil Sci. Soc. Am. J. 46:476-481.

White, I., D. E. Smiles and K. M. Perroux. 1979. "Absorption of Water by Soil: The Constant Flux Boundary Condition." Soil Sci. Soc. Am. J. 43:659-664

APPENDIX A

INFIL1D USER'S MANUAL

APPENDIX A

INFIL1D USER'S MANUAL

The INFIL1D program solves two integral equations, (7) and (10), to define the position of a wetting front as a function of time for a constant-flux boundary condition. Equation (7) defines the depth, z, as an integral function of the water content, and Equation (10) defines the time as an integral function of the surface water content. In order to define the wetting front for a specific time, Equation (10) must be solved interatively for various surface water contents until the value corresponding to specified time is found. The INFIL1D program increments the surface water content between the upper and lower limits and solves for the corresponding times. If specific times are required, the surface water content associated with the specified time is interpolated.

```
INPUT GUIDE FOR PROGRAM INFIL1D
WRITTEN BY TOM MCKEON AND STEVE SIMMONS
LAST REVISION:    12/1982
THIS PROGRAM IS DESIGNED TO CALCULATE WETTING FRONT MOVEMENT
FOR CONSTANT FLUX BOUNDARY CONDITIONS (ONE DIMENSIONAL)
THE INPUT FILE IS  HYDRA.DAT
THE OUTPUT FILES INCLUDE;
1) A LINE PRINTER PLOT CALLED THETA.PLT
2) A LISTING OF CALCULATED VALUES WRITTEN TO UNIT 6
3) A PEN PLOTTER DATA FILE CALLED INFIL.RES , THIS FILE
   IS FORMATTED AS INPUT TO THPLT.FLX WHICH IS CONTROLLED
   BY THPLT.CTL

THE NECESSARY DATA TO RUN INFIL1D ARE:


        VARIABLE NAME          DEFINITION                        FORMAT
        *************          **********                        ******

1          NPTS           THE NUMBER OF DATA POINTS              FREE FORMAT
                          USED TO INTERPOLATE DIFFUSIVITY
                          AND HYDRAULIC CONDUCTIVTY VALUES
```

2	WC(I=1,NPTS)	WATER CONTENT	FREE	FORMAT
3	HC(I=1,NPTS)	HYDRAULIC CONDUCTIVITY	"	"
4	DF(I=1,NPTS)	DIFFUSIVITY	"	"
5	THETAN	THE WATER CONTENT AT TIME=0	"	"
6	THETA1	A SOIL WATER CONTENT ARBITRARILY CLOSE TO THETAN (THIS IS USED TO AVOID SINGULARITIES)	"	"
7	THSAT	SATURATED SOIL WATER CONTENT	"	"
8	VO	THE CONSTANT FLUX IMPOSED AT THE SOIL SURFACE	"	"
9	THLESS	A SMALL NUMBER SUBTRACTED FROM THSAT TO AVOID SINGULARITIES AS THLESS IS DECREASED THE PROGRAM WILL EXAMINE GREATER TIMES	"	"
10	NTIME	NUMBER OF TIME INTERVALS	"	"
11	NSTEP	NUMBER OF SPACE (Z) INTERVALS	"	"
12	AERR	THE MAXIMUM ABSOLUTE ERROR IN THE NUMERICAL INTEGRATION	"	"
13	RERR	THE RELATIVE ERROR IN THE NUMERICAL INTEGRATION (IF RERR=0.01 THEN THE INTEGRATION ROUTINE WILL ATTEMPT TO SOLVE THE INTEGRAL WITHIN AN ACCURACY OF TWO DECIMALS PLACES)	"	"
14	TDEVF	FRACTIONAL TIME DEVIATION; THIS PARAMETER DEFINES THE ACCURACY REQUIRED IN THE INTERPOLATION OF SPECIFIED TIMES	"	"
15	IFLAG1	1 OR 2 IF IFLAG1=1 THE PROBLEM IS AN INFILTRATION WITH GRAVITY FORMULATION IF IFLAG1=2 THE PROBLEM IS FORMULATED AS AN ABSORPTION PROBLEM (GRAVITY FREE)	"	"
16	IFLAG2	1, 2, OR 3 DETERMINES THE TYPE OF APPROXIMATION USED AS THE FUNCTIONAL F	"	"
17	IFLAG3	0 OR 1 DETERMINES THE TYPE OF DATA THAT IS PRINTED TO THE PLOTTER INPUT FILE (INFIL.RES) IF IFLAG3 IS 0 THE ACTUAL CALCULATED DATA POINTS ARE PRINTED IF IFLAG3=1 THEN DATA POINTS ARE I INTERPOLATED TO EQUAL INCREMENTED Z STEPS	"	"
18	M	THE NUMBER OF SPECIFIED TIMES TO BE	"	"

		EXAMINED (IF TDEVF=0.0 THIS IS NOT REQUIRED)	
19	T(I,1=M)	SPECIFIC TIMES TO BE EVALUATED (IF TDEVF=0.0 THIS IS NOT REQUIRED)	FREE FORMAT
20	ZMAX	MAXIMUM DEPTH (USED WHEN IFLAG3=1)	" "
21	NSTEPZ	NUMBER OF EQUAL SPACED INTERVALS TO INTERPOLATE WATER CONTENT VALUES (USED WHEN IFLAG3=1)	" "

This sample input file was used to generate the test case compared with the results of Haverkamp et al. (1977). The results of this test case are shown in Figure 1.

```
23,
.1,.11,.12,.13,.14,.15,.16,.17,.18,.19,.20,.21,.22,.23,.24,.25,.26,.27,.28,
.285,.286,.2864,.287
.13307,0.2121,.30634,.418,.5502,.70711,.89434,1.1194,1.3926,1.7284,2.1472,
2.6794,3.3713,4.2970,5.5814,7.45,10.344,15.218,24.253,31.274,32.761,33.318,
34.0,
93.55,101.94,112.25,124.75,139.92,158.44,181.3,209.9,246.28,293.49,356.28,
442.31,564.64,747.03,1036.4,1536.7,2518.8,4902.3,14244.,45459.,79475.,
.11813E+06,.16100E+06
0.1,0.101,0.287,13.69,0.00001
20,25
0.0,0.001,0.01
1,3,0
4,0.2,0.3,0.4,0.6,
0.,0
```

APPENDIX B

PROGRAM LISTINGS

APPENDIX B

PROGRAM LISTINGS

```
C       PROGRAM        INFIL1D.FLX
C       PROGRAM TO CALCULATE WETTING FRONT MOVEMENT
C       FOR CONSTANT FLUX BOUNDARY CONDITION
C
C                * 11/15/82 *
C
        COMMON /DATA/ WC(100),HC(100),DF(100),NPTS
        COMMON /PAR/ THETAN,HCN,FLUX,THETAO,IFLAG1,IFLAG2,IERROR
        REAL MASBAL
        DIMENSION T(25),GRAF(27,2),ZCOORD(27),THZ(27),H(27),QDAY(27)
        EXTERNAL FNG, FNH, DCFLUX
        FC(THETA)= 1./(THETA-THETAN) - HCN*A/FLUX
        DATA NHOUR,NTOTAL,QINF,QRAIN,QPE,IPLANT,IDAY/1,1,0.,0.,0.,0,0/
C
C       READ HYDRAULIC DATA: WATER CONTENT, CONDUCTIVITY, DIFFUSIVITY
C       & NUMBER OR POINTS.
        OPEN(UNIT=1,NAME='HYDRA.DAT',TYPE='OLD')
        READ(1,*) NPTS
        READ(1,*) (WC(I), I=1,NPTS)
        READ(1,*) (HC(I), I=1,NPTS)
        READ(1,*) (DF(I), I=1,NPTS)
C
C       LOG TRANSFORM DATA
        DO (I=1,NPTS)
        HC(I)=ALOG(HC(I))
        DF(I)=ALOG(DF(I))
        FIN
C
C       READ CONTROL PARAMETERS
        READ(1,*) THETAN,THETA1,THSAT,VO,THLESS
        READ(1,*) NTIME,NSTEP
        READ(1,*) AERR,RERR,TDEVF
        READ(1,*) IFLAG1,IFLAG2,IFLAG3
C       SET DEFAULT VALUES
        IF(THETAN.EQ.0.) THETAN=.001
        IF(THETA1.LE.THETAN) THETA1=1.01*THETAN
        IF(THLESS.EQ.0.) THLESS=THETA1-THETAN
C
C       READ SELECTED OUTPUT TIMES
        WHEN(TDEVF.GT.0.) READ(1,*) M,(T(I),I=1,M)
        ELSE M=1
C       TIME DEVIATION FRACTION TDEVF = 0 GIVES WETTING FRONTS
C       FOR EQUALLY SPACED SURFACE WATER CONTENT STEPS
        READ(1,*)ZMAX,NSTEPZ
        IF(NSTEPZ.GT.25)NSTEPZ=25
        IF(NSTEP.GT.25)NSTEP=25
        CLOSE(UNIT=1)
        OPEN(UNIT=4,NAME='THETA.PLT',TYPE='NEW')
        OPEN(UNIT=3,NAME='INFIL.RES',TYPE='NEW',FORM='UNFORMATTED')
```

```
C
      IF(IFLAG3.EQ.0)ZMAX=0.
C
      IF(ZMAX.EQ.0)NSTEPZ=NSTEP
      NPT=NSTEPZ+1
      NPT0=NSTEP+1
      DTIME=NTIME
      DELZ=ZMAX/NSTEPZ



      DO (I=1,NPT)
      GRAF(I,1)=DELZ*(I-1)
      FIN
      WRITE(3)NPT,IPLANT,(GRAF(LL,1),LL=1,NPT),DTIME

C
C
C     TEST CALCULATE HYDRAULIC PROPERTIES
      DTHETA=(THSAT-THETA1)/NSTEP
      HCN=FNK(THETAN)
      DFN=FND(THETAN)
      IF(IFLAG1.EQ.1)FLUX=VO-HCN
      IF(IFLAG1.EQ.2)FLUX=VO
      THETAO=THSAT
      WRITE(6,*)'HYDRAULIC PROPERTIES:'
      WRITE(6,904)
 904  FORMAT(10X,'STEP',3X,'THETA',7X,'K(THETA)',8X,'D(THETA)')
      WRITE(6,*) NSTEP,THETAN,HCN,DFN
      DO (I=1,NSTEP+1)
      THETA=DTHETA*(I-1)+THETA1
      WRITE(6,*) I,THETA,FNK(THETA),FND(THETA)
      WRITE(6,*) 'FNG(THETA)=',FNG(THETA)
      FIN
C
C     DETERMINE EXPONENTIAL FORM NEAR INITIAL WATER CONTENT
      A=(ALOG(FNK(THETA1))-ALOG(HCN))/(THETA1-THETAN)
      B=(ALOG(FND(THETA1))-ALOG(DFN))/(THETA1-THETAN)
C
C     RESTRICT SURFACE WATER CONTENT RANGE BASED ON FLUX VO
      AMAX=ALOG(VO)
      CALL INTERP(HC,WC,NPTS,AMAX,THMAX)
      WRITE(6,*) ' '
      THMAXO=THMAX-THLESS
      IF(VO.GT.FNK(THSAT))
      THMAXO=THSAT
      IPOND=1
      FIN
      WRITE(6,*)'MAX THETA=',THMAX
      WRITE(6,*)'APPLIED MAX THETA=',THMAXO
```

```
C
C      CHECK FOR METHOD COMPATIBILITY ERRORS AS A RESULT
C      OF SINGULARITIES IN FUNCTION  FNG(THETA)
       WRITE(6,*) ' '
       WRITE(6,*)'METHOD COMPATIBILITY ERRORS'
       IF(FC(THMAX).LE.0.)
       WRITE(6,1)
       ICOUNT=ICOUNT+IERROR
       FIN
       THETA=THETA1
       TEST=FNG(THETA)
       IF(IERROR.EQ.1) WRITE(6,*) THETA,TEST
       ICOUNT=ICOUNT+IERROR
       DO (I=1,NPTS)
       THETA=WC(I)
       IF(THETA.GT.THETA1.AND.THETA.LT.THMAX)
       TEST=FNG(THETA)
       IF(IERROR.EQ.1) WRITE(6,*) THETA,TEST
       ICOUNT=ICOUNT+IERROR
       FIN
       FIN
       THETA=THMAXO
       TEST=FNG(THETA)
       IF(IERROR.EQ.1) WRITE(6,*) THETA,TEST
       ICOUNT=ICOUNT+IERROR
       WHEN(ICOUNT.EQ.0) WRITE(6,*) '*** NO ERRORS DETECTED'
       ELSE
       WRITE(6,*) 'NUMBER OF ERRORS DETECTED=',ICOUNT
       WRITE(6,*) '*** PROGRAM STOP'
       STOP
       FIN
C
C      EVALUATE INTEGRAL OF FNH(THETA) FOR TIMES CORRESPONDING
C      TO VARIOUS SURFACE WATER CONTENTS ,THETAO
       DTHETA=(THMAXO-THETA1)/NTIME
C      CALCULATES INITIAL INTEGRAL FROM ASYMPTOTIC
C      FORM OF FNG(THETA) NEAR THETAN
       FAC=DFN*(EXP(B*(THETA1-THETAN)) - 1.)/B
       SUM1=FAC/FC(THETA1)
       TIME=SUM1/FLUX**2
       T1=TIME
       THETAO=THMAXO
       SUM1=FAC/FC(THETAO)
       SUM2=DCADRE(FNH,THETA1,THETAO,AERR,RERR,ERROR,IER)
       ERRORT=ERROR/FLUX**2
       TP=(SUM1+SUM2)/FLUX**2
       TIMES=TIME
       THETAS=THETA1
       WRITE(6,*)'--------------------------------------------------'
               WRITE(6,*)' F APPROXIMATION USED  TYPE= ',IFLAG2
       WHEN(IFLAG1.EQ.1)WRITE(6,*) '   INFILTRATION PROBLEM '
       ELSE WRITE (6,*)'   ADSORPTION PROBLEM  (GRAVITY FREE)'
       WRITE(6,*)'THETA1=',THETA1,' INITIAL TIME=',T1
       WHEN(IPOND.EQ.1)
       WRITE(6,*)' TIME TO PONDING ',TP
       FIN
```

```
      ELSE
      WRITE(6,*) 'MAXIMUM SIMULATION TIME  ',TP
      FIN
      K=1
      I=0
C
      IF(IPOND.EQ.0)
      Y1=ALOG(THMAXO-THETA1)
      YM=ALOG(THMAX-THMAXO)
      DELTAY=(Y1-YM)/NTIME
      FIN
      DT=(TP-T1)/NTIME
      R=ALOG(TP/T1)/(THMAXO-THETA1)
      R=1./R
C
      TYPE *,'HOW DO YOU WANT THE RESULTS PRESENTED  ??'
      TYPE *,'ENTER (0) FOR REAL Z DIMENSIONS'
      TYPE *,'ENTER (1) FOR TRANSFORMED Z    (Z=Z*(VO-KN))'
      READ(5,*)IZDIM

C
      REPEAT WHILE(I.LT.NTIME.AND.K.LE.M)
      I=I+1
C
C     THIS STATEMENT INCREMENTS THETAO BY UNIFORMLY SPACED INTERVALS
C      IF NO SPECIFIC TIMES ARE TO BE INTERPOLATED
C
      WHEN(TDEVF.EQ.0.)
        THETAO=THETA1+DTHETA*I
      FIN
C
C     THIS SECTION  INCREMENTS  THETAO LOGARITHMICALLY SO THAT POINTS
C     ARE CONCENTRATED NEAR THE UPPER END OF THE THETA RANGE IN ORDER
C     TO FACILITATE BETTER INTERPOLATIONS AT SPECIFIED TIMES
C
      ELSE
      WHEN(IPOND.EQ.1)
      THETAO=THMAXO+R*ALOG((T1+DT*I)/TP)
      IF(THETAO.GT.THMAXO)THETAO=THMAXO
      FIN
C
      ELSE
      YSTEP=Y1-DELTAY*I
      THETAO=THMAX-EXP(YSTEP)
      FIN
      FIN
C

      SUM1=FAC/FC(THETAO)
      SUM2=DCADRE(FNH,THETA1,THETAO,AERR,RERR,ERROR,IER)
      ERRORT=ERROR/FLUX**2
      TIME=(SUM1+SUM2)/FLUX**2
      TTIME=TIME
      TTO=TTIME
      IF(I.EQ.NTIME)
      TTIME=TP
      TTO=TTIME
```

```
      IF(IPOND.EQ.1.AND.T(M).GT.TP)TTO=T(M)
      FIN
      TTHETA=THETAO
C     INTERPOLATE SELECTED TIMES T(K)
      WHILE(TTO.GE.T(K).AND.K.LE.M)
      IF(TDEVF.GT.0.)
      WHEN(T(K).LT.TP)
      TOLD=TIMES
      THOLD=THETAS
      TDEV=TDEVF*T(K)
C     ITERATE INTERPOLATION TO SPECIFIED DEVIATION IN TIME
      REPEAT UNTIL(TTDEV.LT.TDEV.OR.TTDEV.GT.TDEVS.OR.TIME.GT.T(K))
      THOPRE=THETAO
      THETAO=(TTHETA-THOLD)*(T(K)-TTIME)/(TTIME-TOLD)+TTHETA
      SUM1=FAC/FC(THETAO)
      SUM2=DCADRE(FNH,THETA1,THETAO,AERR,RERR,ERROR,IERO)
      TNEW=(SUM1+SUM2)/FLUX**2
      TDEVS=ABS(TIME-T(K))
      TTDEV=ABS(TNEW-T(K))
      WHEN(TTDEV.LE.TDEVS)
      TIME=TNEW
      THOLD=THETAO
      ERRORT=ERROR/FLUX**2
      IER=IERO
      FIN
      ELSE THETAO=THOPRE
      TOLD=TIME
      FIN
      IF(TIME.GT.TP)
      TIME=TP
      THETAO=THMAXO
      FIN
      FIN
      ELSE TIME=T(K)
      FIN
C     CALCULATE WETTING FRONT FOR INTERPOLATED TIME
      IDAY=IDAY+1
      WRITE(6,*)'-----------------------------------------------------'
      WRITE(6,*)'CONSTANT FLUX INFILTRATION RESULTS *** FLUX=',VO
      WRITE(6,*)'WATER CONTENT AT SURFACE:'
      WRITE(6,*)'THETAO=', THETAO,'TIME= ',TIME
      IF(TIME.LE.TP)WRITE(6,*)'ERROR IN TIME=',ERRORT,' IER=',IER
      IF(IERROR.EQ.1) WRITE(6,1)
1     FORMAT(1X,'*** SINGULAR INTEGRAND ERROR')
C
C     CALCULATE WETTING FRONT DEPTHS
      DELTH=(THETAO-THETA1)/NSTEP
      Z=0.
      WRITE(6,1010)
 1010   FORMAT(4X,'THETA',7X,'DEPTH (Z)',7X,'ERROR TERM',5X,'IER NUMBER
     1  ',2X,'DARCY FLUX')
C
      WHEN(TIME .GT.TP)ZSAT=(TIME-TP)*VO/THSAT
      ELSE ZSAT=0.
      GRAF(1,1)=0.
      GRAF(1,2)=THETAO
      ZCOORD(1)=0.
```

```
      THZ(1)=THETAO
      GRAF(2,1)=ZSAT
      GRAF(2,2)=THETAO
      ZCOORD(2)=ZSAT
      THZ(2)=THETAO
      WRITE(6,1011)GRAF(1,2),GRAF(1,1),VO
      WRITE(6,1011)GRAF(2,2),GRAF(2,1),VO
 1011 FORMAT(F10.7,6X,F10.6,28X,F10.6)
      DO (J=1,NSTEP)
      THETAZ=THETAO-DELTH*J
      SUM=DCADRE(FNG,THETAZ,THETAO,AERR,RERR,ERROR,IERA)
      Z=SUM/FLUX+ZSAT
      ERRORZ=ERROR/FLUX
      QDAY(J+2)=DCFLUX(THETAZ)
      GRAF(J+2,1)=Z
      IF(IZDIM.EQ.1)Z=Z*FLUX
      WRITE(6,*) THETAZ,Z,ERRORZ,IERA,QDAY(J+2)
      IF(IERROR.EQ.1) WRITE(6,1)
      GRAF(J+2,2)=THETAZ
      ZCOORD(J+2)=Z
      THZ(J+2)=THETAZ
      FIN
      WRITE(6,*)'------------------------------------------------------------'
      STORAG=(THETA1-THETAN)*GRAF(NPT0+1,1)+ZSAT*THSAT
      DO (L=3,NPT0+1)
      STORAG=STORAG+((GRAF(L,1)+GRAF(L-1,1))/2.-ZSAT)*(THZ(L-1)-THZ(L))
      FIN
      RINPUT=FLUX*TIME
      MASBAL=RINPUT-STORAG
      RPRCNT=MASBAL/RINPUT*100.
      SUM1=FAC/FC(THETAO)
      STAIL=SUM1/FLUX
      NSTP=NSTEP+1
      CALL GRAPH(GRAF,NSTP,TIME,THETAO,THETAN)
      WRITE(6,*)'NET CULMULATIVE FLUX THROUGH THE SYSTEM',RINPUT,
     1  'STORAGE DIFFERENCE IN THE SYSTEM',STORAG
      WRITE(6,*) 'THE ERROR IN WATER BALANCE CALCULATIONS=',MASBAL,
     1        '  % ERROR =',RPRCNT,' STORAGE IN TAIL =',STAIL
        WHEN(ZMAX.GT.0)       DELZ=ZMAX/NSTEPZ
      ELSE DELZ=ZCOORD(NPT0)/NSTEP
      IF(IZDIM.EQ.1)
      DO (LL=1,NPT)
      GRAF(LL,1)=ZCOORD(LL)
      FIN
      FIN
      IF(IFLAG3.EQ.1)
      DO (LL=1,NPT)
      ZSTEP=DELZ*(LL-1)
      CALL INTERP(ZCOORD,THZ,NPT,ZSTEP,THTZ)
      GRAF(LL,1)=ZSTEP
      GRAF(LL,2)=THTZ
      QDAY(LL)=DCFLUX(THTZ)
      FIN
      FIN
      WRITE(3)IDAY,VO,(GRAF(LL,1),GRAF(LL,2),QDAY(LL),LL=1,NPT),
     1  NHOUR,NTOTAL,QINF,QRAIN,QPE
```

```
C       UPDATE TO NEXT SELECTED TIME
        K=K+1
        TIME=TTIME
        THETAO=TTHETA
        FIN
C       UPDATE SAVED PREVIOUS VALUES
        IF(TDEVF.EQ.0.) K=1
        TIMES=TTIME
        THETAS=TTHETA
        FIN
        STOP
        END


C       SUBPROGRAMS FOR INFIL1D.FLX MAIN
C                * 10/22/82 *
C
C       LINEAR INTERPOLATION FOR INCREASING X
        SUBROUTINE INTERP(X,Y,N,Z,W)
        DIMENSION X(N),Y(N)
        K=0
        DO (I=1,N)
        IF(Z.GT.X(I)) K=I
        FIN
        WHEN( K.GT.0.AND.K.LT.N)
        SLOPE=(Y(K+1)-Y(K))/(X(K+1)-X(K))
        W=SLOPE*(Z-X(K)) + Y(K)
        FIN
        ELSE
        IF(K.EQ.0)
        SLOPE=(Y(2)-Y(1))/(X(2)-X(1))
        W=SLOPE*(Z-X(1))+Y(1)
        FIN
        IF(K.EQ.N) W=Y(N)
        FIN
        RETURN
        END
C
C       COMPUTE CONDUCTIVITY BY INTERPOLATION
C       LOG TRANSFORMED VALUES
        FUNCTION FNK(THETA)
        COMMON /DATA/ WC(100),HC(100),DF(100),NPTS
        CALL INTERP(WC,HC,NPTS,THETA,FNK)
        FNK=EXP(FNK)
        RETURN
        END
C
C       COMPUTE DIFFUSIVITY BY INTERPOLATION
C       LOG TRANSFORMED VALUES
        FUNCTION FND(THETA)
        COMMON /DATA/ WC(100),HC(100),DF(100),NPTS
        CALL INTERP(WC,DF,NPTS,THETA,FND)
        FND=EXP(FND)
        RETURN
        END
```

```
C
      FUNCTION FNG(THETA)
      COMMON /PAR/ THETAN,HCN,FLUX,THETAO,IFLAG1,IFLAG2,IERROR
      IERROR=0
      PHI=(THETA-THETAN)/(THETAO-THETAN)
      CONDITIONAL
      (IFLAG2.EQ.1)F=SIN(3.141592/2.*PHI**(3.141592/4.))
      (IFLAG2.EQ.2)F=PHI**(2.-3.141592/4.)
        (IFLAG2.EQ.3)F=PHI
      FIN
      CONDITIONAL
      (IFLAG1.EQ.1)FNG=FND(THETA)/(F-(FNK(THETA)-HCN)/FLUX)
      (IFLAG1.EQ.2)FNG=FND(THETA)/F
      FIN
      IF(FNG.LT.0.) ERROR=1
      RETURN
      END

C
C     COMPUTE THE DARCY FLUX
C
      FUNCTION DCFLUX(THETA)
      COMMON /PAR/ THETAN,HCN,FLUX,THETAO,IFLAG1,IFLAG2,IERROR
      PHI=(THETA-THETAN)/(THETAO-THETAN)
      CONDITIONAL
      (IFLAG2.EQ.1)F=SIN(3.141592/2.*PHI**(3.141592/4.))
      (IFLAG2.EQ.2)F=PHI**(2.-3.141592/4.)
        (IFLAG2.EQ.3)F=PHI
      FIN
      DCFLUX=(FLUX*F)+HCN
      RETURN
      END

      FUNCTION FNH(THETA)
      COMMON /PAR/ THETAN,HCN,FLUX,THETAO,IFLAG1,IFLAG2,IERROR
      FNH=(THETA-THETAN)*FNG(THETA)
      RETURN
      END


      SUBROUTINE GRAPH(GRAF,NSTEP,TIME,THETAO,THETAN)
      DIMENSION GRAF(NSTEP,2)
      DATA DOT,ZIP/'*',0.0/
      WRITE(4,909)THETAO,TIME
 909  FORMAT('1','***',20X,'THETA  VS.  DEPTH (Z)',' THETA0= ',F7.4,
     1  2X,'TIME = ',F12.5,/,6X,'Z','***',1X
     2  '0.0',26X,'0.1',27X,'0.2',27X,'0.3',/,11X,119('-'))
      KASIM=300.*THETAN
      IF(KASIM.EQ.0)KASIM=1
      KTST=40.*(GRAF(1,1)/GRAF(NSTEP,1))
      IF(KTST.GE.2)
      KK=300.*THETAO
      J=KK-KASIM-1
      WRITE(4,24)ZIP,DOT
      FIN
      KZ=0
```

```
      DO (I=1,NSTEP)
      K2=40.*(GRAF(I,1)/GRAF(NSTEP,1))
      KK=300.*(GRAF(I,2))
      J=KK-KASIM-1
      KZ=K2-KZ
      CONDITIONAL
      (KZ.LT.1)CONTINUE
      (KZ.LT.2.AND.J.GE.1)WRITE(4,24)GRAF(I,1),DOT
      (J.LT.1)WRITE(4,26)GRAF(I,1),DOT
      (KZ.GE.2)WRITE(4,25)GRAF(I,1),DOT
      FIN
24    FORMAT(' ',G9.2,'I',<KASIM>(X),'.',<J>(' '),A1)
25    FORMAT(<KZ>(10X,'I',<KASIM>(X),'.',/)
     1  ,' ',G9.2,'I',<KASIM>(X),'.',<J>(' '),A1)
26    FORMAT(<KZ>(10X,'I',<KASIM>(X),'.',/),' ',G9.2,'I',<KASIM>(X)
     1  ,A1)
      KZ=K2
      FIN
      RETURN
      END
```

DISTRIBUTION

No. of Copies	
	OFFSITE
	D. E. Large, National Program Manager ORO Radioactive Waste Management Program Oak Ridge Operations Office U.S. Department of Energy P.O. Box E Oak Ridge, TN 37830
5	L. J. Mezga, Program Manager Low-Level Waste Management Program Oak Ridge National Laboratory P.O. Box X Oak Ridge, TN 37830
	M. J. Barainca, Program Manager Low-Level Waste Management Program Idaho Operations Office U.S. Department of Energy 550 Second Street Idaho Falls, ID 83401
	R. Boland Waste Management Project Office Nevada Operations Office U.S. Department of Energy P. O. Box 14100 Las Vegas, NV 89114
	C. P. Gertz, Chief Radioactive Waste Technology Branch Idaho Operations Office U.S. Department of Energy 550 Second Street Idaho Falls, ID 83401
	J. J. Jicha, Director R&D and Byproducts Division DP-123 (GTN) U.S. Department of Energy Washington, DC 20545
	J. M. McGough, Director Waste Management and Transportation Development Division Albuquerque Operations Office U.S. Department of Energy P.O. Box 5400 Albuquerque, NM 87115
	S. Mann, Senior Program Manager Technical Management Division Chicago Operations Office U.S. Department of Energy 9800 S. Cass Avenue Argonne, IL 60439
	H. Saucier Process and Weapons Division Savannah River Operations Office U.S. Department of Energy P.O. Box A Aiken, SC 29801
27	Technical Information Center U.S. Department of Energy P.O. Box 62 Oak Ridge, TN 37830
	C. A. Abrams, Manager of Radiol. Eng. Argonne National Laboratory - West P.O. Box 2528 Idaho Falls, ID 83401

No. of Copies		No. of Copies	
	Technical Library Argonne National Laboratory Argonne, IL 60439		International Ground Water Modeling Center Holcomb Research Institute Butler University Indianapolis, IN 46208
	B. Rawles Battelle Memorial Institute Office of Nuclear Waste Isolation 505 King Avenue Columbus, OH 43201	2	Technical Library Idaho National Engineering Laboratory Idaho Falls, ID 83401
	P. Columbo, Group Leader Nuclear Waste Research Brookhaven National Laboratory Building 701 Upton, NY 11973		T. Harvey Mail Stop L-262 Lawrence Livermore National Laboratory P.O. Box 808 Livermore, CA 94550
	R. H. Beers, Manager Waste Management Programs Division EG&G Idaho, Inc. P.O. Box 1625 Idaho Falls, ID 83415		D. T. Oakley Program Manager for Waste Management Los Alamos National Laboratory P.O. Box 1663 Los Alamos, NM 87545
	Program Manager Low-Level Waste Management Program EG&G Idaho, Inc. P.O. Box 1625 Idaho Falls, Idaho 83415		J. G. Steger Group Leader Environmental Science (LS-6) MS K-495 Los Alamos National Laboratory P.O. Box 1663 Los Alamos, NM 87544
	G. L. Meyer Project Leader LLW Standard U.S. Environmental Protection Agency Office of Radiation Programs ANR-460 Washington, DC 20460	2	Technical Library Los Alamos National Laboratory Los Alamos, NM 87545
	Environmental Protection Agency Technology Assessment Division Office of Radiation Programs Washington, DC 20460		National Academy of Sciences National Research Council 2101 Constitution Avenue Washington, DC 20418
			E. F. Conti, Chief Waste Management Branch U.S. Nuclear Regulatory Commission Mail Stop 1130-SS Washington, DC 20555

No. of Copies	
	E. O'Donnell U.S. Nuclear Regulatory Commission Mail Stop 1130-SS Washington, DC 20555
2	NRC Division of Technical Information and Document Control U.S. Nuclear Regulatory Commission Washington, DC 20555
2	Technical Library Oak Ridge National Laboratory Oak Ridge, TN 37830
	E. L. Albenesuis Savannah River Laboratory P.O. Box A Aiken, SC 29801
2	Technical Library Savannah River National Laboratory Aiken, SC 29801
	J. Fischer Low-Level Radioactive Waste Program U.S. Geological Survey Water Resources Division 12201 Sunrise Valley Drive Reston, VA 22092
	V. C. Rogers Rogers & Associates Engineering Corporation P.O. Box 330 Salt Lake City, UT 84110

No. of Copies	
ONSITE	
	Hanford Engineering Development Laboratory
	R. E. Lerch
2	Richland Operations Office
	M. Dayoni H. E. Ransom
3	Rockwell Hanford Operations
	J. F. Albaugh J. W. Patterson J. L. Deichman
64	Pacific Northwest Laboratory
	C. R. Cole D. W. Dragnich M. D. Freshley G. W. Gee (10) P. C. Hays L. T. Jones C. T. Kincaid L. T. Lakey T. J. McKeon R. W. Nelson A. E. Reisenauer C. S. Simmons (35) S. W. Tyler L. L. Wendell Technical Information (5) Publishing Coordination (2)